

İleri Yöneylem Araştırması Uygulamaları GAMS

Dr. Özgür Kabak

GAMS

- ▶ Matematiksel programlama modellerini çözmek için bir platform
- ▶ Doğrusal, doğrusal olmayan, tam sayılı vs. modellerin çözümü için kullanılabilir.



GAMS

- ▶ Kümeler (sets)
- ▶ Parametreler (parameters, Tables)
- ▶ Karar değişkenleri (variables, positive variables, integer variables, binary variables)
- ▶ Kısıtlar (equations)



Örnek Problem

Sabit maliyetli ulaştırma modeli

- ▶ ATK- Beyaz dört müşterisinin taleplerini karşılamak için üç potansiyel fabrikayı değerlendirmektedir. Her hangi bir fabrikayı açarsa firma 200,000TL'lik yatırım yapması gerekmektedir.
- ▶ Bir fabrikada üretilip, bir müşteriye gönderme birim maliyetleri tabloda verilmiştir.
- ▶ Firmanın talepleri en düşük maliyetle karşılaması için gerekli TP modelini kurunuz.

	Müşteri 1	Müşteri 2	Müşteri 3	Müşteri 4	Kapasite
Fabrika 1	8TL	6TL	10TL	9TL	60,000
Fabrika 2	9TL	12TL	13TL	7TL	55,000
Fabrika 3	14TL	9TL	16TL	5TL	45,000
Talepler	40,000	30,000	25,000	15,000	

Sabit maliyetli ulařtırma modeli - TP

► Karar deęiřkenleri

- x_{ij} : i fabrikasından j müşteriğine gönderilen miktar ($i = 1,2,3; j = 1,2,3,4$)
- y_i : i fabrikasının açılması ($i = 1,2,3$) $y_i = \begin{cases} 1 & i \text{ fabrikası açılırsa} \\ 0 & \text{açılmazsa} \end{cases}$

► TP modeli

$$\text{Min } \sum_i \sum_j c_{ij} x_{ij} + 200000 \sum_i y_i$$

$$\text{Öyle ki; } \sum_j x_{ij} \leq k_i y_i \quad (i = 1,2,\dots,m) \quad \text{kapasite kısıtları}$$

$$\sum_i x_{ij} \geq t_j \quad (j = 1,2,\dots,n) \quad \text{Talep kısıtları}$$

$$x_{ij} \geq 0, y_i \in \begin{Bmatrix} 0 \\ 1 \end{Bmatrix}$$



Ulaştırma modeli

```
sets i santraller /1, 2, 3/  
     j şehirler /1*4/ ;
```

```
table c(i,j)
```

	1	2	3	4
1	8	6	10	9
2	9	12	13	7
3	14	9	16	5

```
;
```

```
parameter talep(j) /1 40, 2 30, 3 25, 4 15 /;  
parameter arz(i) / 1 40, 2 45, 3 25 /;
```

```
positive variables x(i,j);  
variable z;
```

```
equations amac, arzkisiti, talepkisiti ;
```

```
amac.. z =e= sum((i,j), x(i,j)*c(i,j));  
arzkisiti(i).. sum(j, x(i,j)) =l= arz(i);  
talepkisiti(j).. sum(i, x(i,j)) =g= talep(j);
```

```
model ulastirma /all/;  
solve ulastirma using LP minimizing z;  
display x.l;
```




GAMS UYGULAMASI

Maximize f

$$\text{Subject to } \sum_{j=1}^m x_{ij} - \sum_{k=1}^m x_{ki} = \begin{cases} f & \text{if } i = 1 \\ 0 & \text{if } i \neq 1 \text{ or } m \\ -f & \text{if } i = m \end{cases}$$

$$x_{ij} \leq u_{ij} \quad i, j = 1, 2, \dots, m$$

$$x_{ij} \geq 0 \quad i, j = 1, 2, \dots, m$$



```

sets i /1*7/;
sets n(i) /2*6/;
alias (i,j);
sets ag(i,j) /1.2, 1.3, 2.3 , 2.4 , 2.5 , 3.4 , 3.6, 4.5,
4.6, 5.7, 6.5, 6.7 /;

```

```

parameter u(i,j)

```

```

/ 1.2 6
  1.3 7
  2.3 1
  2.4 3
  2.5 4
  3.4 2
  3.6 5
  4.5 1
  4.6 2
  5.7 5
  6.5 2
  6.7 4 /;

```

```

positive variables x(i,j);

```

```

variable f;

```

```

equations e1, e2, e3, e4;

```

```

e1.. sum(j, x("1",j)) =e= f;

```

```

e2(n).. sum(j$ag(n,j), x(n,j)) - sum(i$ag(i,n), x(i,n))=e= 0;

```

```

e3.. sum(i, x(i,"7"))=e= f;

```

```

e4(i,j)$ag(i,j).. x(i,j) =l= u(i,j);

```

```

model eba /all/;

```

```

solve eba using LP max f;

```

```

display x.l;

```

► $\$, \text{Ord}(i)$

► $\sum_{i>j} x_{ij} \geq 1 \quad \forall j$

`denk1em(j).. sum(i$(ord(i)>ord(j)),x(i,j))=g=1;`

► $x(i) > y(j) \quad \forall j > i + 1$

`Denk1em(i,j)$(ord(j)>ord(i)+1).. xi(i) =g= y(j);`

Veri Alma (Küme&Tablo)

- Problemin çözümü için gerekli tüm verileri GAMS'in içine yazmak yerine, bu verileri saklayan başka bir dosya(excel, notepad) kullanılabilir. Bu sayede hem kod basitleşir hem de var olan verileri tekrar yazmaya gerek kalmaz
- Importing sets&table from an excel file

	A	B	C	D	E	F
1		m1	m2	m3	m4	
2	f1	8	6	10	9	
3	f2	9	12	13	7	
4	f3	14	9	16	5	
5						

Veri Alma (Küme&Tablo)

SET Kümeadı /

\$call =xls2gms r=ilkhücre:sonhücre i=dosyaadı.xls o=setKümeadı.inc

\$include setKümeadı.inc

/;

```
sets i fabrikalar /  
$call =xls2gms r=a2:a4 i=data.xls o=seti.inc  
$include seti.inc  
/;
```

- ▶ GAMS sadece dikey verileri okuyabilir, yatay verilerin okunabilmesi için koda ilave bir terim eklenmelidir

SET Kümeadı /

\$call =xls2gms r= ilkhücre:sonhücre s="," i=dosyaadı.xls o=setKümeadı.inc

\$include set Kümeadı.inc

/;

```
sets      j müşteriler /  
$call =xls2gms r=b1:e1 s="," i=data.xls o=setj.inc  
$include setj.inc  
/;
```

Veri Alma (Küme&Tablo)

► Tabloları okumak için;

TABLE **tabloadı**(1.küme,2.küme)

\$call =xls2gms r=**hücre**(0,0):**hücre**(n,m) i=**dosyaadı**.xls o=part**tabloadı**.inc

\$include part**tabloadı**.inc

;

```
TABLE c(i,j)
$call =xls2gms r=a1:e4 i=data.xls o=parc.inc
$include parc.inc
;
```

	A	B	C	D	E	F
1		m1	m2	m3	m4	
2	f1	8	6	10	9	
3	f2	9	12	13	7	
4	f3	14	9	16	5	
5						

```

sets i fabrikalar /
$call =xls2gms r=a2:a4 i=data.xls o=seti.inc
$include seti.inc
/;
sets      j müşteriler /
$call =xls2gms r=b1:e1 s="," i=data.xls o=setj.inc
$include setj.inc
/;
TABLE c(i,j)
$call =xls2gms r=a1:e4 i=data.xls o=parc.inc
$include parc.inc
;
scalar fa /200000/;
parameter talep(j) /m1 40000, m2 30000, m3 25000, m4 15000 /;
parameter kapasite(i) / f1 60000, f2 55000, f3 45000 /;

positive variables x(i,j);
binary variables y(i);
variable z;

equations amac, kapasitekisiti, talepkisiti ;

amac.. z =e= sum((i,j), x(i,j)*c(i,j)) + fa*sum(i, y(i));
kapasitekisiti(i).. sum(j, x(i,j)) =l= kapasite(i)*y(i);
talepkisiti(j).. sum(i, x(i,j)) =g= talep(j);

model ulastirma /all/;
solve ulastirma using MIP minimizing z;
display x.l, y.l;

```

